Representations of A_5 using GAP*

David Joyner

March 4, 2003

Abstract

We show how to use some commands in GAP [GAP] to investigate finite group representations. In particular, we study the representations of A_5. This note is written for the first-time user of GAP but assumes that the reader knows basic things about groups and representations.

Contents

*Originally written 8-2000. Slightly revised 3-2003.

1 Introduction

Very briefly, in this note we shall first learn how to enter a finite group, how to list their elements, their conjugacy classes, and their centralizers. Then we shall study their representations, more precisely their characters [1]. We shall also study the induced representations $ind_H^G\ \chi$, where H is a certain subgroup of G, and G is either A_5 or its 2-fold cover $SL(2,5)$. We shall discuss A_5 from the following points of view:

- as a permutation group,
- as a matrix group ($SL(2,4)$),
- as the quotient of a matrix group ($PSL(2,5)$),
- as a finitely presented group ($\langle\, a,b \mid a^2 = b^3 = (ab)^5 = 1\rangle$).

GAP [GAP] is a free software package designed originally for groups but which now does much more. GAP's home is now at the Mathematical Institute, St. Andrews, Scotland (homepage: `http://www.gap-system.org/`), where it is managed by (among others) Steve Linton. Other workers (there are too many to list here) are listed at the above URL under the authors link. There is also a manual (both in latex and in html), an email list for bugs, one for GAP questions, and installation instructions, all listed on the GAP homepage.

Though this is written for the GAP novice, it will be assumed that you have GAP up and running. Sometimes a GAP command will be given but the GAP output will not be given (though quite often we give both), since

[1]Until the last section, we shall assume all the word "representation" refers to a finite dimensional complex representation. The last section briefly discusses modular representations.

it will be assumed you can see it yourself. Many commands assume that the commands previously discussed have already been entered.

For the background on groups and representations, we refer to Serre [S].

2 Getting started

Once you start GAP, you will see a banner like this:

```
            #########           ######         ###########           ###
         #############          ######         ############         ####
        ##############         ########        #############       #####
       ###############         ########        #####   ######      #####
      ######         #         #########       #####    #####     ######
     ######                   ##########       #####    #####    #######
     #####                   ##### ####        #####   ######   ########
     ####                    #####  #####      #############   ###  ####
     #####     #######       ####    ####      ###########    ####  ####
     #####     #######      #####    #####     ######        ####   ####
     #####     #######      #####    #####     #####         #############
      #####      #####     ################    #####         #############
      ######     #####     ################    #####         #############
      ################    ##################   #####                ####
       ###############    #####        #####   #####                ####
         #############    #####        #####   #####                ####
          #########      #####          #####  #####                ####

     Information at:  http://www-gap.dcs.st-and.ac.uk/~gap
     ? for help. Copyright and authors list by ?copyright, ?authors

   Loading the library. Please be patient, this may take a while.
GAP4, Version: 4.1 fix 7 of 11-Nov-1999, i686-pc-linux-gnu-gcc
Components:  small, small2, small3, id2, id3, trans, prim, tbl,
             tom  installed.
```

From this you already know I am running linux on by pentium PC and have started gap, version 4.1.

I recommend you log every session. To do this, type

`gap> LogTo("/home/wdj/gapfiles/altern_gp.log");`

for example (we will be doing an example based on the alternating group A_5, so the file name will be called `altern_gp.log`). You don't type " `gap>`", by the way, as that is simply the GAP cursor beginning each line.

3 Entering groups

To begin, let us enter the alternating group into GAP by typing

`gap>A5:=AlternatingGroup(5);`

You can see by typing `Size(A5);` that it has 360 elements. Its elements are obtained by typing

`gap>a5:=Elements(A5);`

Since $A_5 \cong PSL(2, \mathbb{F}_4)$, you can also enter this group into GAP by the command

```
gap>PSL24:=PSL(2,4);
```

if you prefer. This will also be a permutation group in GAP. On the other hand, if you enter $SL(2, \mathbb{F}_4)$ using

```
gap>SL24:=SL(2,4);
```

(which is the same group after all, since we're in characteristic 2), you will get a group of *matrices*, not a permutation group, whose elements are written in the form

```
[ [ [ 0*Z(2), Z(2)^0 ], [ Z(2)^0, 0*Z(2) ] ],
  [ <an immutable GF2 vector of length 2>, <an immutable GF2 vector of length 2> ],
  [ [ 0*Z(2), Z(2)^0 ], [ Z(2)^0, Z(2^2) ] ],
  [ [ 0*Z(2), Z(2)^0 ], [ Z(2)^0, Z(2^2)^2 ] ],
  [ [ 0*Z(2), Z(2^2) ], [ Z(2^2)^2, 0*Z(2) ] ],
  [ [ 0*Z(2), Z(2^2) ], [ Z(2^2)^2, Z(2)^0 ] ],
  [ [ 0*Z(2), Z(2^2) ], [ Z(2^2)^2, Z(2^2) ] ],
  [ [ 0*Z(2), Z(2^2) ], [ Z(2^2)^2, Z(2^2)^2 ] ],
  [ [ 0*Z(2), Z(2^2)^2 ], [ Z(2^2), 0*Z(2) ] ],
  [ [ 0*Z(2), Z(2^2)^2 ], [ Z(2^2), Z(2)^0 ] ],
  [ [ 0*Z(2), Z(2^2)^2 ], [ Z(2^2), Z(2^2) ] ],
  [ [ 0*Z(2), Z(2^2)^2 ], [ Z(2^2), Z(2^2)^2 ] ],
  [ [ Z(2)^0, 0*Z(2) ], [ 0*Z(2), Z(2)^0 ] ],
  [ [ Z(2)^0, 0*Z(2) ], [ Z(2)^0, Z(2)^0 ] ],
  [ [ Z(2)^0, 0*Z(2) ], [ Z(2^2), Z(2)^0 ] ],
  [ [ Z(2)^0, 0*Z(2) ], [ Z(2^2)^2, Z(2)^0 ] ],
  [ [ Z(2)^0, Z(2)^0 ], [ 0*Z(2), Z(2)^0 ] ],
  [ <an immutable GF2 vector of length 2>, <an immutable GF2 vector of length 2> ],
  [ [ Z(2)^0, Z(2)^0 ], [ Z(2^2), Z(2^2)^2 ] ],
  [ [ Z(2)^0, Z(2)^0 ], [ Z(2^2)^2, Z(2^2) ] ],
  [ [ Z(2)^0, Z(2^2) ], [ 0*Z(2), Z(2)^0 ] ],
  [ [ Z(2)^0, Z(2^2) ], [ Z(2)^0, Z(2^2)^2 ] ],
  [ [ Z(2)^0, Z(2^2) ], [ Z(2^2), Z(2^2) ] ],
  [ [ Z(2)^0, Z(2^2) ], [ Z(2^2)^2, 0*Z(2) ] ],
  [ [ Z(2)^0, Z(2^2)^2 ], [ 0*Z(2), Z(2)^0 ] ],
  [ [ Z(2)^0, Z(2^2)^2 ], [ Z(2)^0, Z(2^2) ] ],
  [ [ Z(2)^0, Z(2^2)^2 ], [ Z(2^2), 0*Z(2) ] ],
  [ [ Z(2)^0, Z(2^2)^2 ], [ Z(2^2)^2, Z(2^2)^2 ] ],
 [ [ Z(2^2), 0*Z(2) ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2), 0*Z(2) ], [ Z(2)^0, Z(2^2)^2 ] ],
  [ [ Z(2^2), 0*Z(2) ], [ Z(2^2), Z(2^2)^2 ] ],
  [ [ Z(2^2), 0*Z(2) ], [ Z(2^2)^2, Z(2^2)^2 ] ],
  [ [ Z(2^2), Z(2)^0 ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2), Z(2)^0 ], [ Z(2)^0, 0*Z(2) ] ],
  [ [ Z(2^2), Z(2)^0 ], [ Z(2^2), Z(2^2) ] ],
  [ [ Z(2^2), Z(2)^0 ], [ Z(2^2)^2, Z(2)^0 ] ],
  [ [ Z(2^2), Z(2^2) ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2), Z(2^2) ], [ Z(2)^0, Z(2^2) ] ],
  [ [ Z(2^2), Z(2^2) ], [ Z(2^2), Z(2)^0 ] ],
  [ [ Z(2^2), Z(2^2) ], [ Z(2^2)^2, 0*Z(2) ] ],
  [ [ Z(2^2), Z(2^2)^2 ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2), Z(2^2)^2 ], [ Z(2)^0, Z(2)^0 ] ],
  [ [ Z(2^2), Z(2^2)^2 ], [ Z(2^2), 0*Z(2) ] ],
  [ [ Z(2^2), Z(2^2)^2 ], [ Z(2^2)^2, Z(2^2) ] ],
  [ [ Z(2^2)^2, 0*Z(2) ], [ 0*Z(2), Z(2^2) ] ],
  [ [ Z(2^2)^2, 0*Z(2) ], [ Z(2)^0, Z(2^2) ] ],
  [ [ Z(2^2)^2, 0*Z(2) ], [ Z(2^2), Z(2^2) ] ],
  [ [ Z(2^2)^2, 0*Z(2) ], [ Z(2^2)^2, Z(2^2) ] ],
  [ [ Z(2^2)^2, Z(2)^0 ], [ 0*Z(2), Z(2^2) ] ],
  [ [ Z(2^2)^2, Z(2)^0 ], [ Z(2)^0, 0*Z(2) ] ],
  [ [ Z(2^2)^2, Z(2)^0 ], [ Z(2^2), Z(2)^0 ] ],
  [ [ Z(2^2)^2, Z(2)^0 ], [ Z(2^2)^2, Z(2^2)^2 ] ],
  [ [ Z(2^2)^2, Z(2^2) ], [ 0*Z(2), Z(2^2) ] ],
  [ [ Z(2^2)^2, Z(2^2) ], [ Z(2)^0, Z(2)^0 ] ],
  [ [ Z(2^2)^2, Z(2^2) ], [ Z(2^2), Z(2^2)^2 ] ],
  [ [ Z(2^2)^2, Z(2^2) ], [ Z(2^2)^2, 0*Z(2) ] ],
  [ [ Z(2^2)^2, Z(2^2)^2 ], [ 0*Z(2), Z(2^2) ] ],
  [ [ Z(2^2)^2, Z(2^2)^2 ], [ Z(2)^0, Z(2^2)^2 ] ],
```

```
[ [ Z(2^2)^2, Z(2^2)^2 ], [ Z(2^2), 0*Z(2) ] ],
[ [ Z(2^2)^2, Z(2^2)^2 ], [ Z(2^2)^2, Z(2)^0 ] ] ]
```

(I don't know why GAP occasionally says "an immutable GF2 vector of length 2" in place of the actual matrix; possibily it is due to memory restrictions of the particular machine.) You may also enter this group as

```
gap>PSL25:=PSL(2,5);
```

if you wish. It is also a permutation group in GAP and by typing

```
gap>Elements(PSL25);
```

you find its elements are

```
[ (), (3,5)(4,6), (2,3)(4,5), (2,3,4,6,5), (2,4)(5,6), (2,4,5,3,6),
  (2,5,6,4,3), (2,5)(3,6), (2,6)(3,4), (2,6,3,5,4), (1,2)(4,6), (1,2)(3,5),
  (1,2,3)(4,5,6), (1,2,3,6,5), (1,2,4)(3,5,6), (1,2,4,3,6), (1,2,5,4,3),
  (1,2,5)(3,4,6), (1,2,6,5,4), (1,2,6)(3,4,5), (1,3,2)(4,6,5), (1,3,4,5,2),
  (1,3)(5,6), (1,3,6,4,5), (1,3)(2,4), (1,3,5)(2,4,6), (1,3,4)(2,5,6),
  (1,3,6)(2,5,4), (1,3,2,6,4), (1,3,5,2,6), (1,4,5,6,2), (1,4,2)(3,6,5),
  (1,4,3,5,6), (1,4)(3,6), (1,4,6,2,3), (1,4,2,3,5), (1,4,6)(2,5,3),
  (1,4)(2,5), (1,4,3)(2,6,5), (1,4,5)(2,6,3), (1,5,6,3,2), (1,5,2)(3,6,4),
  (1,5,4,6,3), (1,5)(3,4), (1,5,6)(2,3,4), (1,5,4)(2,3,6), (1,5,3,2,4),
  (1,5,2,4,6), (1,5,3)(2,6,4), (1,5)(2,6), (1,6,3,4,2), (1,6,2)(3,5,4),
  (1,6)(4,5), (1,6,5,3,4), (1,6)(2,3), (1,6,4)(2,3,5), (1,6,3)(2,4,5),
  (1,6,5)(2,4,3), (1,6,2,5,3), (1,6,4,2,5) ]
```

All these groups are known to be isomorphic (probably a proof can be found in [R] but I'm not sure). One way to see this using GAP is to type

```
gap>IsomorphismTypeFiniteSimpleGroup(A5);
gap>IsomorphismTypeFiniteSimpleGroup(PSL25);
gap>IsomorphismTypeFiniteSimpleGroup(SL24);
```

They all have the same type. Alternatively, you can type

```
gap>IsSimple(A5); Size(A5);
gap>IsSimple(PSL25); Size(PSL25);
gap>IsSimple(SL24); Size(SL24);
```

They are all simple groups of size 60, so by the classification of simple groups they must be isomorphic!

Next, we compute all the conjugacy classes of A_5:

```
gap> A5_classes:=ConjugacyClasses(A5);
```

GAP will reply with something like

```
[ ()^G, (1,2)(3,4)^G, (1,2,3)^G, (1,2,3,4,5)^G, (1,2,3,5,4)^G ]
```

This is a list of classes. To get their representatives in A_5, you can type

```
Representative(A5_classes[1]);
```

for example. We shall not do so here since it is easy enough in this case to simply type in an element from each class. Now let us find all the non-trivial centralizers in A_5, up to conjugacy. Type

```
gap> C1:=Centralizer(A5,(1,2)(3,4)); Size(C1);
gap> C2:=Centralizer(A5,(1,2,3)); Size(C2);
gap> C3:=Centralizer(A5,(1,2,3,4,5)); Size(C3);
gap> C4:=Centralizer(A5,(1,2,3,5,4)); Size(C4);
```

GAP returns the groups, at least the groups in GAP's notation, and their sizes $(4, 3, 5, 5$, resp.). Another way is to type

```
gap> cent_a5:=List(A5_classes,x->Centralizer(A5,Representative(x)));
```

Up to conjugation, there are no other centralizers in A_5. To find the elements in C_2 for example, type

```
gap> Elements(C2);
```

(or `gap> Elements(cent_a5[2]);`). GAP will reply with something like

```
[ (), (1,2,3), (1,3,2) ]
```

Next, we compute their normalizers in A_5, along with their sizes:

```
gap> N1:=Normalizer(A5,C1); Size(N1);
Group([ (1,2)(3,4), (1,3)(2,4), (2,3,4) ])
12
gap> N2:=Normalizer(A5,C2); Size(N2);
Group([ (1,2,3), (2,3)(4,5) ])
6
gap> N3:=Normalizer(A5,C3); Size(N3);
Group([ (1,2,3,4,5), (2,5)(3,4) ])
10
gap> N4:=Normalizer(A5,C4); Size(N4);
Group([ (1,2,3,5,4), (2,4)(3,5) ])
10
```

By typing

```
gap> IsAbelian(C1);,
```

to which GAP replies `true`, we find out that C_1 is abelian. (It a group of size 4, so of course it is abelian!) Similarly, we find that all the centralizers are abelian (so A_5 is a commutative transitive group) and all the normalizers are non-abelian.

Now, it turns out that any commutative transitive group G has the following property: if $C, C' \subset G$ are any two centralizers then either $C \cap C' = \{1\}$ (i.e., are "disjoint") or $C = C'$. To check this property for $G = A_5$ and $C = C_1$, it suffices to type

```
gap>L:=List(a5,x->Size(Intersection(C1,C1^x)));
gap>L:=List(a5,x->Size(Intersection(C1,C2^x)));
gap>L:=List(a5,x->Size(Intersection(C1,C3^x)));
gap>L:=List(a5,x->Size(Intersection(C1,C4^x)));
```

All the elements of each list L end up being either 1 (if their intersection is trivial) or 4 (if they are equal). For example, to see which x satisfies $|C_1 \cap C_1^x| = 4$, type

```
for i in [1..Size(A5)] do if L[i]=4 then Print(a5[i],''\n''); fi; od;
```

Note that the output is precisely the elements of N_1, as expected.

Type

```
gap> R1:=RightCosets(N1,C1);
[ RightCoset(Group( [ (1,2)(3,4), (1,3)(2,4) ] ),()),
  RightCoset(Group( [ (1,2)(3,4), (1,3)(2,4) ] ),(2,3,4)),
  RightCoset(Group( [ (1,2)(3,4), (1,3)(2,4) ] ),(2,4,3)) ]
gap> Representative(R1[1]);
()
gap> Representative(R1[2]);
(2,3,4)
gap> Representative(R1[3]);
(2,4,3)
```

to get the coset representatives of

$$C_1 = \{(), (1,2)(3,4), (1,3)(2,4), (1,4)(2,3)\}$$

in

$$N_1 = \{(), (2,3,4), (2,4,3), (1,2)(3,4), (1,2,3), (1,2,4), (1,3,2), \\ (1,3,4), (1,3)(2,4), (1,4,2), (1,4,3), (1,4)(2,3)\}.$$

Thus the representatives of N_1/C_1 are $\{(), (2,3,4), (2,4,3)\}$.

Remark 1. If instead you type

```
gap> CosetTableBySubgroup(N1,C1);
```

then GAP returns

```
[ [ 1, 2, 3 ], [ 1, 2, 3 ], [ 1, 2, 3 ], [ 1, 2, 3 ], [ 2, 3, 1 ], [ 3, 1, 2 ] ],
```

What does this mean? Recall N_1 has generators $(1,2)(3,4), (1,3)(2,4), (2,3,4)$. (If you don't believe me, type

```
gap> GeneratorsOfGroup(N1);
```

which will get GAP to tell you this.) For each generator and its inverse, GAP computes its effect on the cosets N_1/C_1 as a permutation. There are $3 = |N_1/C_1|$ cosets, so the permutation is in S_3, in this case. The permutations outputted by the `CosetTableBySubgroup` command are written in row notation (so $[1,2,3]$ is the identity, for example). In particular, we find that the `CosetTableBySubgroup` command tells us that the permutation representation of N_1 acting on N_1/C_1 is isomorphic to a cyclic subgroup of S_3.

4 Representations of A_5

Now we turn to representation theory. Since the centralizers are abelian in this case, all its irreducible representations are 1-dimensional. For the set of all irreducible representations of C_1, you may type either

```
gap> C1_chars:=LinearCharacters(C1);
```

(which returns the linear, i.e. degree 1, irreducible characters of the group), or `gap> C1_chars:=Irr(C1);`
(which returns all the irreducible characters of the group). You can see the output (which is GAP's notation for a character of C_1) is the same in either case. Let us abbreviate the characters of C_1 by

$$\mu_{1,1}\ (=1), \quad \mu_{1,2}, \quad \mu_{1,3}, \quad \mu_{1,4}.$$

The set of irreducible characters of C_1 is denoted C_1^*.

We obtain

```
gap> C1_chars:=Irr(C1);
[ Character( CharacterTable( Group([ (1,2)(3,4), (1,3)(2,4) ]) ),
    [ 1, 1, 1, 1 ] ), Character( CharacterTable( Group([ (1,2)(3,4), (1,3)(2,4)
     ]) ), [ 1, 1, -1, -1 ] ), Character( CharacterTable( Group(
    [ (1,2)(3,4), (1,3)(2,4) ]) ), [ 1, -1, 1, -1 ] ),
  Character( CharacterTable( Group([ (1,2)(3,4), (1,3)(2,4) ]) ),
    [ 1, -1, -1, 1 ] ) ]
gap> C2_chars:=Irr(C2);
[ Character( CharacterTable( Alt( [ 1 .. 3 ] ) ), [ 1, 1, 1 ] ),
  Character( CharacterTable( Alt( [ 1 .. 3 ] ) ), [ 1, E(3), E(3)^2 ] ),
  Character( CharacterTable( Alt( [ 1 .. 3 ] ) ), [ 1, E(3)^2, E(3) ] ) ]
gap> C3_chars:=Irr(C3);
[ Character( CharacterTable( Group([ (1,2,3,4,5) ]) ), [ 1, 1, 1, 1, 1 ] ),
  Character( CharacterTable( Group([ (1,2,3,4,5) ]) ),
    [ 1, E(5), E(5)^2, E(5)^3, E(5)^4 ] ), Character( CharacterTable( Group(
    [ (1,2,3,4,5) ]) ), [ 1, E(5)^2, E(5)^4, E(5), E(5)^3 ] ),
  Character( CharacterTable( Group([ (1,2,3,4,5) ]) ),
    [ 1, E(5)^3, E(5), E(5)^4, E(5)^2 ] ), Character( CharacterTable( Group(
    [ (1,2,3,4,5) ]) ), [ 1, E(5)^4, E(5)^3, E(5)^2, E(5) ] ) ]
gap> C4_chars:=Irr(C4);
[ Character( CharacterTable( Group([ (1,2,3,5,4) ]) ), [ 1, 1, 1, 1, 1 ] ),
  Character( CharacterTable( Group([ (1,2,3,5,4) ]) ),
    [ 1, E(5), E(5)^2, E(5)^4, E(5)^3 ] ), Character( CharacterTable( Group(
    [ (1,2,3,5,4) ]) ), [ 1, E(5)^2, E(5)^4, E(5)^3, E(5) ] ),
  Character( CharacterTable( Group([ (1,2,3,5,4) ]) ),
    [ 1, E(5)^3, E(5), E(5)^2, E(5)^4 ] ), Character( CharacterTable( Group(
    [ (1,2,3,5,4) ]) ), [ 1, E(5)^4, E(5)^3, E(5), E(5)^2 ] ) ]
```

What does this mean? First, we find the conjugacy classes of each group $C_1, ..., C_4$. Since each of these groups is abelian, the conjugacy classes correspond to the elements of the groups themselves:

```
gap> C1_classes:=ConjugacyClasses(C1);
[ ()^G, (1,2)(3,4)^G, (1,3)(2,4)^G, (1,4)(2,3)^G ]
gap> C2_classes:=ConjugacyClasses(C2);
[ ()^G, (1,2,3)^G, (1,3,2)^G ]
gap> C3_classes:=ConjugacyClasses(C3);
[ ()^G, (1,2,3,4,5)^G, (1,3,5,2,4)^G, (1,4,2,5,3)^G, (1,5,4,3,2)^G ]
gap> C4_classes:=ConjugacyClasses(C4);
[ ()^G, (1,2,3,5,4)^G, (1,3,4,2,5)^G, (1,4,5,3,2)^G, (1,5,2,4,3)^G ]
```

Consider for example the character

```
Character( CharacterTable( Group([ (1,2)(3,4), (1,3)(2,4) ]) ), [ 1, 1, 1, 1 ] )
```

of C_1 which we shall denote by μ. This is a homomophism $\mu : C_1 \to \mathbb{C}^\times$. The last entry `[ 1, 1, 1, 1 ]` indicates the values of this character μ on the classes

`[ ()^G, (1,2)(3,4)^G, (1,3)(2,4)^G, (1,4)(2,3)^G ]`.

For example, $\mu((1,4)(2,3)) = 1$. Consider for example the character

```
Character( CharacterTable( Group( [ (1,2,3,5,4) ]) ), [ 1, E(5)^4, E(5)^3, E(5), E(5)^2 ] )
```

of C_4 which we shall denote by μ'. The last entry

`[ 1, E(5)^4, E(5)^3, E(5), E(5)^2 ]`

indicates the values of this character μ on the classes

`[ ()^G, (1,2,3,5,4)^G, (1,3,4,2,5)^G, (1,4,5,3,2)^G, (1,5,2,4,3)^G ]`.
For example, $\mu((1,3,4,2,5)) = \zeta_5^3$, where ζ_5 is a primitive 5^{th} root of unity (denoted `E(5)` is GAP).

A question which arises later (in the next section) when we induce these characters to A_5 is the following. Is the character

```
Character( CharacterTable( Group([ (1,2)(3,4), (1,3)(2,4) ]) ), [ 1, 1, -1, -1 ] )
```

of C_1, call it μ'', invariant under the action of the normalizer?

We know that the coset representatives of N_1/C_1 are $\{(), (2,3,4), (2,4,3)\}$. GAP computes $(2,3,4)^{-1}C_1(2,3,4)$ using the command

```
(2,3,4)^(-1)*Elements(C1)*(2,3,4);
```

to be `[ (), (1,3)(2,4), (1,4)(2,3), (1,2)(3,4) ]`, where `Elements(C1)` is `[ (), (1,2)(3,4), (1,3)(2,4), (1,4)(2,3) ]`. We see that this will change the values `[ 1, 1, -1, -1 ]` of μ'' to `[ 1, -1, -1, 1 ]` for the values of $(\mu'')^{(2,3,4)}$. Therefore, the answer to the question is no, the character μ'' is not invariant under N_1. In fact, since N_1/C_1 is cyclic we find that $(\mu'')^{(2,4,3)} \neq \mu''$, so the stabilizer of μ'' in N_1/C_1 is trivial. We call such a character μ'' "regular".

Let us abbreviate the characters of C_2 by

$$\mu_{2,1}\ (=1), \quad \mu_{2,2}, \quad \mu_{2,3},$$

the characters of C_3 by

$$\mu_{3,1}\ (=1), \quad \mu_{3,2}, \quad \mu_{3,3}, \quad \mu_{3,4}, \quad \mu_{3,5},$$

and the characters of C_4 by

$$\mu_{4,1}\ (=1), \quad \mu_{4,2}, \quad \mu_{4,3}, \quad \mu_{4,4}, \quad \mu_{4,5},$$

4.1 Character values for A_5

The group A_5 has 5 conjugacy classes:

$$\{1\}, \quad \{a = (1,2)(3,4)\}, \quad \{b = (1,2,3)\},$$

$$\{c = (1,2,3,4,5)\}, \quad \{d = (1,3,5,2,4)\}.$$

There are 5 irreducible characters. Their values on the conjugacy classes are summarized as follows:

tr $\pi\backslash\lvert\{\gamma\}\rvert$	1	15	20	12	12
tr $\pi\backslash\{\gamma\}$	$\{1\}$	$\{a\}$	$\{b\}$	$\{c\}$	$\{d\}$
tr π_1	1	1	1	1	1
tr π_2	3	-1	0	ϕ	$\overline{\phi}$
tr π_3	3	-1	0	$\overline{\phi}$	ϕ
tr π_4	4	0	1	-1	-1
tr π_5	5	1	-1	0	0

Here $\phi = (1+\sqrt{5})/2$ and $\overline{\phi} = (1-\sqrt{5})/2$.

The representations of A_5 are obtained by typing

```
A5_repns:=Irr(A5);
```

GAP will reply

```
[ Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 1, 1, 1, 1, 1 ] ),
  Character( CharacterTable( Alt( [ 1 .. 5 ] ) ),
    [ 3, -1, 0, -E(5)^2-E(5)^3, -E(5)-E(5)^4 ] ),
  Character( CharacterTable( Alt( [ 1 .. 5 ] ) ),
    [ 3, -1, 0, -E(5)-E(5)^4, -E(5)^2-E(5)^3 ] ),
  Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 4, 0, 1, -1, -1 ] ),
  Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 5, 1, -1, 0, 0 ] ) ]
```

Let us abbreviate them by

$$\pi_1\ (=1),\quad \pi_2,\quad \pi_3,\quad \pi_4,\quad \pi_5.$$

The set of irreducible characters of A_5 is denoted A_5^*. If you type

```
gap> Display(CharacterTable(A5));
```

then GAP will return

```
CT1

     2  2  2  .  .  .
     3  1  .  1  .  .
     5  1  .  .  1  1

       1a 2a 3a 5a 5b
    2P 1a 1a 3a 5b 5a
    3P 1a 2a 1a 5b 5a
```

```
     5P 1a 2a 3a 1a 1a

X.1      1  1  1  1  1
X.2      3 -1  .  A *A
X.3      3 -1  . *A  A
X.4      4  .  1 -1 -1
X.5      5  1 -1  .  .

A = -E(5)^2-E(5)^3
  = (1+ER(5))/2 = 1+b5
```

The notation of the conjugacy classes agrees with that of the ATLAS [Atlas]. In particular, the class of the identity element is the first one; thus the degree of a character is the character value in column 1. The other values depend on the ordering of the conjugacy classes (in `A5_classes`) and on the ordering of the irreducible characters (in `A5_repns`).

The individual values of the characters are obtained using the following function[2]:

```
gap> value_char:=function(x,i)
> local pi,j,C;
>  pi:=A5_repns[i];
> C:=ConjugacyClasses(A5);
> for j in [1..Length(C)] do if x in C[j] then return ValuesOfClassFunction(pi)[j]; fi; od;
> end;
```

The command `value_char(x,i)` returns the value of tr $\pi_i(x)$. For example,

```
gap> value_char((),5);
5
gap> value_char((),4);
4
gap> value_char((1,2,3),4);
1
gap> value_char((1,2,3,4,5),4);
-1
gap> value_char((1,2)(3,5),4);
0
```

[2]For those familiar with MAPLE, the similarities between the two is more than coincidental, as the GAP programming language was originally based on MAPLE's.

4.2 Induction

Now, let us induce the first (trivial) character of C_1 from C_1 to A_5:

```
gap> ind1:=InducedClassFunction(C1_chars[1],A5);
```

To study reducibility of representations [3] using GAP, define the **(Schur) scalar product** of a class function χ with a class function ψ on a finite group G by

$$(\chi, \psi) = \frac{1}{|G|} \sum_{g \in G} \chi(g)\psi(g^{-1}).$$

Recall, the scalar product of a character (i.e., the trace of a possibly reducible representation of G) with itself is 1 if and only if it is irreducible.

Is this induced representation irreducible? Type either

```
gap> IsIrreducibleCharacter(ind1);
```

(to which GAP returns `false`) or

```
gap> ScalarProduct(ind1,ind1);
```

(to which GAP returns 6). So, we now know $ind_{C_1}^{A_5} 1$ is not irreducible. What is its decomposition into irreducibles?

$$ind_{C_1}^{A_5} 1 \cong \sum_{\pi \in A_5^*} m(\pi)\pi, \quad m(\pi) \in \mathbb{Z}.$$

To determine the multiplicities $m(\pi)$ it suffices (thanks to Schur orthogonality) to type

```
m:=List(A5_repns,x->ScalarProduct(ind1,x));
```

GAP returns `[ 1, 0, 0, 1, 2 ]`, so, in the notation of 4.1,

$$ind_{C_1}^{A_5} \mu_{1,1} = ind_{C_1}^{A_5} 1 \cong \pi_1 + \pi_4 + 2\pi_5.$$

Likewise, by typing

```
gap> ind2:=InducedClassFunction(C1_chars[2],A5);
Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 15, -1, 0, 0, 0 ] )
gap> ScalarProduct(ind2,ind2);
4
gap> ind3:=InducedClassFunction(C1_chars[3],A5);
Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 15, -1, 0, 0, 0 ] )
```

[3] More precisely, reducibility of characters of representations.

```
gap> ScalarProduct(ind3,ind3);
4
gap> ind4:=InducedClassFunction(C1_chars[4],A5);
Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 15, -1, 0, 0, 0 ] )
gap> ScalarProduct(ind4,ind4);
4
```

we find that none of the induced representations $ind_{C_1}^{A_5}\mu$ are irreducible. Indeed, GAP gives

$$ind_{C_1}^{A_5}\mu_{1,2} \cong ind_{C_1}^{A_5}\mu_{1,3} \cong ind_{C_1}^{A_5}\mu_{1,4} \cong \pi_2 + \pi_3 + \pi_4 + \pi_5.$$

By typing

```
gap> ind1:=InducedClassFunction(C2_chars[1],A5);
Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 20, 0, 2, 0, 0 ] )
gap> ind2:=InducedClassFunction(C2_chars[2],A5);
Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 20, 0, -1, 0, 0 ] )
gap> ind3:=InducedClassFunction(C2_chars[3],A5);
Character( CharacterTable( Alt( [ 1 .. 5 ] ) ), [ 20, 0, -1, 0, 0 ] )
gap> ScalarProduct(ind1,ind1);
8
gap> ScalarProduct(ind2,ind2);
7
gap> ScalarProduct(ind3,ind3);
7
```

we find that none of the induced representations $ind_{C_2}^{A_5}\mu$ are irreducible. Indeed, by entering commands similar to those explained above, GAP tells us that

$$ind_{C_2}^{A_5}\mu_{2,1} = ind_{C_2}^{A_5}1 \cong \pi_1 + \pi_2 + \pi_3 + 2\pi_4 + \pi_5,$$
$$ind_{C_2}^{A_5}\mu_{2,2} \cong ind_{C_2}^{A_5}\mu_{2,3} \cong \pi_2 + \pi_3 + \pi_4 + 2\pi_5.$$

Likewise, we find that none of the induced representations $ind_{C_3}^{A_5}\mu$ or $ind_{C_4}^{A_5}\mu$ are irreducible either. Indeed, by entering commands similar to those explained above, GAP tells us that

$$ind_{C_3}^{A_5}\mu_{3,1} = ind_{C_3}^{A_5}1 \cong ind_{C_4}^{A_5}\mu_{4,1} = ind_{C_4}^{A_5}1 \cong \pi_1 + \pi_2 + \pi_3 + \pi_5,$$
$$ind_{C_3}^{A_5}\mu_{3,2} = ind_{C_3}^{A_5}\mu_{3,5} \cong ind_{C_4}^{A_5}\mu_{4,3} = ind_{C_4}^{A_5}\mu_{4,4} \cong \pi_2 + \pi_4 + \pi_5,$$
$$ind_{C_3}^{A_5}\mu_{3,3} = ind_{C_3}^{A_5}\mu_{3,4} \cong ind_{C_4}^{A_5}\mu_{4,2} = ind_{C_4}^{A_5}\mu_{4,5} \cong \pi_3 + \pi_4 + \pi_5.$$

Remark 2. *Another way to determine whether or not these induced representations (or any other ones you might happen to run across) are irreducible is simply to compare their values with those given by the character table of A_5 as given in §4.1. The Frobenius formula for the character of an induced representation in this case becomes particularly simple.*

Lemma 1. *Let G be a finite commutation transitive group, C a centralizer in G, $\mu \in C^*$ a character, and let $\pi = ind_C^G \mu$. Then*

$$\operatorname{tr} \pi(g) = \begin{cases} \frac{|G|}{|C|}, & g = 1 \\ \frac{1}{|C|} \sum_{n \in N_G(C)} \mu(n^{-1}cn), & c \in Conj(g,G) \cap C \neq \emptyset,\ g \neq 1 \\ 0, & Conj(g,G) \cap C = \emptyset,\ g \neq 1. \end{cases}$$

5 Orbital integrals

To compute orbital integrals in gap, you must write a program (called a "function" in GAP).

```
gap> orbital_integral:=function(g)
> local x,y;
> y:=List(Elements(A5),x->f(x^(-1)*g*x)/Size(A5));
> return Sum(y);
> end;
```

Here f is an as yet undefined function on A_5. First, type

```
gap> class_fcn_A5:=function(x,y)
>  if x in ConjugacyClass(A5,y) then return 1; fi;
> return 0;
>  end;
```

Now type

```
gap> f:=function(x)
> return(class_fcn_A5(x,(1,2)(3,5))+3*class_fcn_A5(x,(1,2,3))-class_fcn_A5(x,()));
> end;
```

The orbital integral of this function on A_5 is computed by typing
`gap> orbital_integral((1,2,3,4));`
to which GAP responds 0 (as it must, since $(1,2,3,4) \notin A_5$). If you typed
`gap> orbital_integral((1,2,3));`
GAP would give you 3, as expected.

6 Principal series of $SL(2,4)$

Here we wish to use GAP to examine which (if any) characters of the standard Borel subgroup $B = \{\begin{pmatrix} * & * \\ 0 & * \end{pmatrix}$ in $G = SL(2,4)$ induce irreducibly to G.

As was indicated already, we may enter $SL(2,4)$, as a matrix group, into GAP by typing

```
gap> SL24:=SL(2,4);
```

One may also enter this using `PSL(2,4)`, which has the advantage that GAP recognizes this as a permutation group [4], but then we would have the problem of determining which subgroup of G the Borel is. To find out which group os the Borel subgroup of this matrix group, let us first list all the entries whose lower leftc orner entry is a 0:

```
gap> L0:=[];
[  ]
gap> for i in [1..60] do if Elements(SL24)[i][2][1]=0*Z(2) then L0:=Concatenation(L0,[i]); fi; od;
gap> Borel:=List(L0,i->Elements(SL24)[i]);
gap> Borel:=List(L0,i->Elements(SL24)[i]);
[ [ [ Z(2)^0, 0*Z(2) ], [ 0*Z(2), Z(2)^0 ] ],
  [ [ Z(2)^0, Z(2)^0 ], [ 0*Z(2), Z(2)^0 ] ],
  [ [ Z(2)^0, Z(2^2) ], [ 0*Z(2), Z(2)^0 ] ],
  [ [ Z(2)^0, Z(2^2)^2 ], [ 0*Z(2), Z(2)^0 ] ],
  [ [ Z(2^2), 0*Z(2) ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2), Z(2)^0 ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2), Z(2^2) ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2), Z(2^2)^2 ], [ 0*Z(2), Z(2^2)^2 ] ],
  [ [ Z(2^2)^2, 0*Z(2) ], [ 0*Z(2), Z(2^2) ] ],
  [ [ Z(2^2)^2, Z(2)^0 ], [ 0*Z(2), Z(2^2) ] ],
  [ [ Z(2^2)^2, Z(2^2) ], [ 0*Z(2), Z(2^2) ] ],
  [ [ Z(2^2)^2, Z(2^2)^2 ], [ 0*Z(2), Z(2^2) ] ] ]
gap> B:=Group(Borel);
<matrix group with 12 generators>
```

We wish to have GAP embed the matrix group $SL(2,4)$ into a permutation group and then use this same embedding for B_0. For this, type

```
gap> iso:=IsomorphismPermGroup(SL24);
<action homomorphism>
gap> B:=Image(iso,B0);
<permutation group of size 5 with 12 generators>
gap> G:=Image(iso,SL24);
Group([ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
  ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ])
```

This is $SL(2,4)$ and its standard Borel as permutation groups.

[4] This is useful since GAP's `Irr` and `LinearCharacters` commands only work for permutation groups.

6.1 Representations of $SL(2,4)$

The irreducible representations of B and G are obtained by typing

```
gap> B_repns:=Irr(BO);
[ Character( CharacterTable( <permutation group of size 12 with
    12 generators> ), [ 1, 1, 1, 1 ] ),
  Character( CharacterTable( <permutation group of size 12 with
    12 generators> ), [ 1, 1, E(3)^2, E(3) ] ),
  Character( CharacterTable( <permutation group of size 12 with
    12 generators> ), [ 1, 1, E(3), E(3)^2 ] ),
  Character( CharacterTable( <permutation group of size 12 with
    12 generators> ), [ 3, -1, 0, 0 ] ) ]
gap> G_repns:=Irr(G);
[ Character( CharacterTable( Group(
    [ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
      ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ), [ 1, 1, 1, 1, 1
     ] ), Character( CharacterTable( Group(
    [ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
      ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ),
    [ 3, -1, 0, -E(5)^2-E(5)^3, -E(5)-E(5)^4 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
      ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ),
    [ 3, -1, 0, -E(5)-E(5)^4, -E(5)^2-E(5)^3 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
      ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ),
    [ 4, 0, 1, -1, -1 ] ), Character( CharacterTable( Group(
    [ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
      ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ), [ 5, 1, -1, 0, 0
     ] ) ]
```

We abbreviate these irreducible representations of B by

$$\psi_1\ (=1), \quad \psi_2, \quad \psi_3, \quad \psi_4.$$

Note ψ_1, ψ_2, ψ_3 are characters of A extended trivially to B. We may abbreviate these irreducible representations of G by

$$\pi_1'\ (=1), \quad \pi_2', \quad \pi_3', \quad \pi_4', \quad \pi_5'.$$

To relate these representations to those obtained in the previous section, we look at the conjugacy classes of G.

```
gap> A_classes:=ConjugacyClasses(A);
[ ()^G, ( 2, 3, 4)( 5,13, 9)( 6,15,12)( 7,16,10)( 8,14,11)^G,
  ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14)^G ]
gap> B_classes:=ConjugacyClasses(B);
[ ()^G, ( 5, 6)( 7, 8)( 9,11)(10,12)(13,16)(14,15)^G,
  ( 2, 3, 4)( 5,13, 9)( 6,15,12)( 7,16,10)( 8,14,11)^G,
  ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14)^G ]
gap> G_classes:=ConjugacyClasses(G);
[ ()^G, ( 5, 6)( 7, 8)( 9,11)(10,12)(13,16)(14,15)^G,
  ( 2, 3, 4)( 5,13, 9)( 6,15,12)( 7,16,10)( 8,14,11)^G,
  ( 2, 5,10,11, 7)( 3, 9,15,16,12)( 4,13, 8, 6,14)^G,
  ( 2, 5,14,16, 8)( 3, 9, 7, 6,10)( 4,13,12,11,15)^G ]
gap> g_classes:=List(G_classes,x->Representative(x));
[ (), ( 5, 6)( 7, 8)( 9,11)(10,12)(13,16)(14,15),
  ( 2, 3, 4)( 5,13, 9)( 6,15,12)( 7,16,10)( 8,14,11),
  ( 2, 5,10,11, 7)( 3, 9,15,16,12)( 4,13, 8, 6,14),
  ( 2, 5,14,16, 8)( 3, 9, 7, 6,10)( 4,13,12,11,15) ]
gap> List(g_classes,x->Order(x));
[ 1, 2, 3, 5, 5 ]
```

Note that the split torus A has one less G-conjugacy class than B,

```
gap> A_classes_in_G:=List(A_classes,x->ConjugacyClass(G,Representative(x)));
[ ()^G, ( 2, 3, 4)( 5,13, 9)( 6,15,12)( 7,16,10)( 8,14,11)^G,
  ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14)^G ]
gap> B_classes_in_G:=List(B_classes,x->ConjugacyClass(G,Representative(x)));
[ ()^G, ( 5, 6)( 7, 8)( 9,11)(10,12)(13,16)(14,15)^G,
  ( 2, 3, 4)( 5,13, 9)( 6,15,12)( 7,16,10)( 8,14,11)^G,
  ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14)^G ]
```

(Note
`gap>( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14) in G_classes[3];`
returns `true`, so only 2 G-conjugacy classes meet A and only 3 G-conjugacy classes meet B.) Let us match up conjugacy classes in the "obvious way",

class of A_5	class of $SL(2,4)$
1	1
$\{(1,2)(3,4)\}$	$\{(5,6)(7,8)(9,11)(10,12)(13,16)(14,15)\}$
$\{(1,2,3)\}$	$\{(2,3,4)(5,13,9)(6,15,12)(7,16,10)(8,14,11)\}$
$\{(1,2,3,4,5)\}$	$\{(2,5,10,11,7)(3,9,15,16,12)(4,13,8,6,14)\}$
$\{(1,2,3,5,4)\}$	$\{(2,5,14,16,8)(3,9,7,6,10)(4,13,12,11,15)\}$

By comparing the character values of these π_i''s with those in the previous section, we find that under the above correspondence of conjugacy classes, π_i matches with π_i' in the sense that their characters are equal on the corresponding classes, $i = 1, 2, 3, 4, 5$.

6.2 Induced representations of $SL(2,4)$

The induced representations from B to G are obtained by typing

```
gap> rho1:=InducedClassFunction(B_repns[1],G);
Character( CharacterTable( Group(
[ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
  ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ), [ 5, 1, 2, 0, 0 ] )
gap> rho2:=InducedClassFunction(B_repns[2],G);
Character( CharacterTable( Group(
[ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
  ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ), [ 5, 1, -1, 0, 0 ] )
gap> rho3:=InducedClassFunction(B_repns[3],G);
Character( CharacterTable( Group(
[ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
  ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ), [ 5, 1, -1, 0, 0 ] )
gap> rho4:=InducedClassFunction(B_repns[4],G);
Character( CharacterTable( Group(
[ ( 2, 4, 3)( 5, 9,13)( 6,12,15)( 7,10,16)( 8,11,14),
  ( 2, 5, 6)( 3, 9,11)( 4,13,16)( 7,14,12)( 8,10,15) ]) ), [ 15, -1, 0, 0, 0 ] )
```

To determine their reducibility, type

```
gap> m1:=List(G_repns,x->ScalarProduct(rho1,x));
 [ 1, 0, 0, 1, 0 ]
gap> m2:=List(G_repns,x->ScalarProduct(rho2,x));
 [ 0, 0, 0, 0, 1 ]
gap> m3:=List(G_repns,x->ScalarProduct(rho3,x));
 [ 0, 0, 0, 0, 1 ]
gap> m4:=List(G_repns,x->ScalarProduct(rho4,x));
 [ 0, 1, 1, 1, 1 ]
```

This tells us that $ind_B^G\ 1$ is reducible. The irreducible representations $ind_B^G\ \psi_2$, $ind_B^G\ \psi_3$ are the "principal series" representations. It also tells us that the induced representation $ind_B^G\ \psi_4$ is reducible and vanishes on the "regular hyperbolic set" $A - \{1\}$.

7 Representations of $PSL(2,5)$

We shall first work with $SL(2,5)$ then mod out by its center.

Type

```
gap> SL25:=SL(2,5);
```

and then type

```
gap> Z_SL25:=Center(SL25);
```

to enter its center. As we know, $PSL(2,5) \cong A_5$.

How do you enter $SL(2,5)/Z$ (or any quotient group, for that matter) into GAP? It is easy if you know the right command. First, you must have a group and a normal subgroup. Now construct the quotient homomorphism of them by typing

```
gap> hom := NaturalHomomorphismByNormalSubgroup(SL25,Z_SL25);
```

The quotient group $PSL(2,5)$ is simply the image under this quotient map:

```
gap> G:=Image(hom);
```

It is just that easy. The elements are

```
gap> psl25:=Elements(G);
[ (), ( 3, 5,12, 9, 8)( 4, 6,11,10, 7), ( 3, 8, 9,12, 5)( 4, 7,10,11, 6),
  ( 3, 9, 5, 8,12)( 4,10, 6, 7,11), ( 3,12, 8, 5, 9)( 4,11, 7, 6,10),
  ( 1, 2)( 3, 4)( 5, 7)( 6, 8)( 9,11)(10,12),
  ( 1, 2)( 3, 6)( 4, 5)( 7,12)( 8,11)( 9,10),
  ( 1, 2)( 3, 7)( 4, 8)( 5,10)( 6, 9)(11,12),
  ( 1, 2)( 3,10)( 4, 9)( 5,11)( 6,12)( 7, 8),
  ( 1, 2)( 3,11)( 4,12)( 5, 6)( 7, 9)( 8,10),
  ( 1, 3)( 2, 4)( 5, 8)( 6, 7)( 9,10)(11,12),
  ( 1, 3, 5)( 2, 4, 6)( 7, 9,11)( 8,10,12),
  ( 1, 3, 8)( 2, 4, 7)( 5,11, 9)( 6,12,10), ( 1, 3,10, 7,12)( 2, 4, 9, 8,11),
  ( 1, 3,11, 6, 9)( 2, 4,12, 5,10), ( 1, 4)( 2, 3)( 5, 6)( 7, 8)( 9,12)(10,11),
  ( 1, 4, 5, 9, 7)( 2, 3, 6,10, 8), ( 1, 4, 8,12, 6)( 2, 3, 7,11, 5),
  ( 1, 4,10)( 2, 3, 9)( 5,12, 7)( 6,11, 8),
```

```
  ( 1, 4,11)( 2, 3,12)( 5, 7,10)( 6, 8, 9),
  ( 1, 5, 3)( 2, 6, 4)( 7,11, 9)( 8,12,10), ( 1, 5,10,11, 8)( 2, 6, 9,12, 7),
  ( 1, 5,12)( 2, 6,11)( 3, 7, 9)( 4, 8,10), ( 1, 5, 7, 4, 9)( 2, 6, 8, 3,10),
  ( 1, 5)( 2, 6)( 3,12)( 4,11)( 7, 8)( 9,10), ( 1, 6,12, 8, 4)( 2, 5,11, 7, 3),
  ( 1, 6)( 2, 5)( 3, 4)( 7,10)( 8, 9)(11,12),
  ( 1, 6,10)( 2, 5, 9)( 3, 8,11)( 4, 7,12), ( 1, 6, 3, 9,11)( 2, 5, 4,10,12),
  ( 1, 6, 7)( 2, 5, 8)( 3,11,10)( 4,12, 9), ( 1, 7, 9, 5, 4)( 2, 8,10, 6, 3),
  ( 1, 7)( 2, 8)( 3, 4)( 5,12)( 6,11)( 9,10),
  ( 1, 7,11)( 2, 8,12)( 3, 5,10)( 4, 6, 9),
  ( 1, 7, 6)( 2, 8, 5)( 3,10,11)( 4, 9,12), ( 1, 7, 3,12,10)( 2, 8, 4,11, 9),
  ( 1, 8, 3)( 2, 7, 4)( 5, 9,11)( 6,10,12), ( 1, 8,11,10, 5)( 2, 7,12, 9, 6),
  ( 1, 8, 9)( 2, 7,10)( 3, 6,12)( 4, 5,11), ( 1, 8)( 2, 7)( 3, 9)( 4,10)( 5, 6)
    (11,12), ( 1, 8, 6, 4,12)( 2, 7, 5, 3,11), ( 1, 9, 6,11, 3)( 2,10, 5,12, 4),
  ( 1, 9)( 2,10)( 3, 4)( 5, 6)( 7,11)( 8,12),
  ( 1, 9,12)( 2,10,11)( 3, 6, 7)( 4, 5, 8), ( 1, 9, 4, 7, 5)( 2,10, 3, 8, 6),
  ( 1, 9, 8)( 2,10, 7)( 3,12, 6)( 4,11, 5),
  ( 1,10, 4)( 2, 9, 3)( 5, 7,12)( 6, 8,11), ( 1,10, 8, 5,11)( 2, 9, 7, 6,12),
  ( 1,10)( 2, 9)( 3, 5)( 4, 6)( 7, 8)(11,12), ( 1,10,12, 3, 7)( 2, 9,11, 4, 8),
  ( 1,10, 6)( 2, 9, 5)( 3,11, 8)( 4,12, 7),
  ( 1,11, 4)( 2,12, 3)( 5,10, 7)( 6, 9, 8), ( 1,11, 5, 8,10)( 2,12, 6, 7, 9),
  ( 1,11, 9, 3, 6)( 2,12,10, 4, 5), ( 1,11)( 2,12)( 3, 8)( 4, 7)( 5, 6)( 9,10),
  ( 1,11, 7)( 2,12, 8)( 3,10, 5)( 4, 9, 6), ( 1,12, 7,10, 3)( 2,11, 8, 9, 4),
  ( 1,12)( 2,11)( 3, 4)( 5, 9)( 6,10)( 7, 8), ( 1,12, 4, 6, 8)( 2,11, 3, 5, 7),
  ( 1,12, 9)( 2,11,10)( 3, 7, 6)( 4, 8, 5),
  ( 1,12, 5)( 2,11, 6)( 3, 9, 7)( 4,10, 8) ]
```

Let us find the standard Borel subgroup of $SL(2,5)$. We use commands analogous to the $SL(2,4)$ case in the previous section.

```
gap>  borel_list:=[];
[  ]
gap> for i in [1..60] do if sl25[i][2][1]=0*Z(5) then borel_list:=Concatenation(borel_list,[i]); fi; od;
gap> borel_list;
[ 21, 26, 31, 36, 41, 46, 51, 56 ]
gap> Borel:=List(borel_list,i->sl25[i]);
[ [ [ Z(5)^0, 0*Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^0 ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^2 ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^3 ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5), 0*Z(5) ], [ 0*Z(5), Z(5)^3 ] ],
  [ [ Z(5), Z(5)^0 ], [ 0*Z(5), Z(5)^3 ] ],
  [ [ Z(5), Z(5) ], [ 0*Z(5), Z(5)^3 ] ] ]
```

This is the list of the elements of the Borel subgroup. The group itself, and the split torus it contains, are constructed by typing

```
gap> B0:=Group(Borel);
<matrix group with 8 generators>
gap> Torus:=[Borel[1],Borel[6]];
[ [ [ Z(5)^0, 0*Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5), 0*Z(5) ], [ 0*Z(5), Z(5)^3 ] ] ]
gap>  A0:=Group(Torus);
Group([ [ [ Z(5)^0, 0*Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5), 0*Z(5) ], [ 0*Z(5), Z(5)^3 ] ] ])
```

These are of size 20 and 4, resp. [5].

The "unipotent radical" is constructed using the commands

[5]In particular, although $SL(2,5)\cong PSL(2,5)$, the standard Borel of $SL(2,4)$ is *not* isomorphic to the standard Borel of $PSL(2,5)$.

```
gap> Nilradical:=[Borel[1],Borel[2],Borel[3],Borel[4],Borel[5]];
[ [ [ Z(5)^0, 0*Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^0 ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^2 ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^3 ], [ 0*Z(5), Z(5)^0 ] ] ]
gap> N0:=Group(Nilradical);
Group([ [ [ Z(5)^0, 0*Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^0 ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5) ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^2 ], [ 0*Z(5), Z(5)^0 ] ],
  [ [ Z(5)^0, Z(5)^3 ], [ 0*Z(5), Z(5)^0 ] ] ])
```

By typing

`Elements(Normalizer(SL25,N0));`

you verify the well-known fact that the normalizer of N_0 in $SL(2,5)$ is B_0.

The respective images of B_0, A_0, and N_0 in $PSL(2,5)$ are obtained by typing `Image(hom,B0);`, `Image(hom,A0);`, `Image(hom,N0);`. It turns out that all the representations induced from a character of the Borel of $PSL(2,5)$ to $PSL(2,5)$ are reducible. For this reason, we instead consider the representations induced from a character of the Borel of $SL(2,5)$ to $SL(2,5)$. Some (half, actually) of these representations factor through to $PSL(2,5)$, as we shall see.

7.1 Principal series of $SL(2,5)$

To obtain the representation of $SL(2,5)$ using GAP, we must first convert it to a permutation group. To this end, type

```
gap> iso:=IsomorphismPermGroup(SL25);
<action homomorphism>
gap> B:=Image(iso,B0);
<permutation group of size 20 with 8 generators>
gap> A:=Image(iso,A0);
Group([ (), ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
    (10,14,18,22) ])
gap> G:=Image(iso,SL25);
Group([ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
    (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)( 8,12,13)
    (14,18,19)(20,24,25) ])
```

The “split torus” A has 4 elements, to be expected since $A \cong \mathbb{F}_5^\times$ is order 4. The irreducible representations of the Borel B are obtained by typing

```
[ Character( CharacterTable( <permutation group of size 20 with
    8 generators> ), [ 1, 1, 1, 1, 1, 1, 1, 1 ] ),
  Character( CharacterTable( <permutation group of size 20 with
    8 generators> ), [ 1, 1, 1, -1, 1, 1, 1, -1 ] ),
  Character( CharacterTable( <permutation group of size 20 with
    8 generators> ), [ 1, 1, 1, -E(4), -1, -1, -1, E(4) ] ),
  Character( CharacterTable( <permutation group of size 20 with
    8 generators> ), [ 1, 1, 1, E(4), -1, -1, -1, -E(4) ] ),
  Character( CharacterTable( <permutation group of size 20 with
    8 generators> ), [ 2, E(5)^2+E(5)^3, E(5)+E(5)^4, 0, 2, E(5)^2+E(5)^3,
      E(5)+E(5)^4, 0 ] ), Character( CharacterTable( <permutation group of size
    20 with 8 generators> ), [ 2, E(5)^2+E(5)^3, E(5)+E(5)^4, 0, -2,
      -E(5)^2-E(5)^3, -E(5)-E(5)^4, 0 ] ),
  Character( CharacterTable( <permutation group of size 20 with
    8 generators> ), [ 2, E(5)+E(5)^4, E(5)^2+E(5)^3, 0, 2, E(5)+E(5)^4,
      E(5)^2+E(5)^3, 0 ] ),
  Character( CharacterTable( <permutation group of size 20 with
    8 generators> ), [ 2, E(5)+E(5)^4, E(5)^2+E(5)^3, 0, -2, -E(5)-E(5)^4,
      -E(5)^2-E(5)^3, 0 ] ) ]
```

Recall that the value of the character on the identity conjugacy class is always the dimension of the representation. Note that the last four representations are not 1-dimensional!

The irreducible representations of the Borel G are obtained by typing

```
gap> G_repns:=Irr(G);
[ Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 1, 1, 1, 1, 1, 1, 1, 1, 1 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 2, E(5)^2+E(5)^3, E(5)+E(5)^4,
      0, -2, -E(5)^2-E(5)^3, -E(5)-E(5)^4, 1, -1 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 2, E(5)+E(5)^4, E(5)^2+E(5)^3,
      0, -2, -E(5)-E(5)^4, -E(5)^2-E(5)^3, 1, -1 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 3, -E(5)^2-E(5)^3, -E(5)-E(5)^4,
      -1, 3, -E(5)^2-E(5)^3, -E(5)-E(5)^4, 0, 0 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 3, -E(5)-E(5)^4, -E(5)^2-E(5)^3,
      -1, 3, -E(5)-E(5)^4, -E(5)^2-E(5)^3, 0, 0 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 4, -1, -1, 0, 4, -1, -1, 1, 1 ] )
    , Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 4, -1, -1, 0, -4, 1, 1, -1, 1 ] )
    , Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 5, 0, 0, 1, 5, 0, 0, -1, -1 ] ),
  Character( CharacterTable( Group(
    [ ( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)
        (10,14,18,22), ( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)
        ( 8,12,13)(14,18,19)(20,24,25) ]) ), [ 6, 1, 1, 0, -6, -1, -1, 0, 0 ] )
 ]
```

To compute the induced representaitons and their multiplicities, type

```
gap> rho:=[0,0,0,0,0,0,0,0];;
gap> m:=[0,0,0,0,0,0,0,0];;
gap> for i in [1..8] do rho[i]:=InducedClassFunction(B_repns[i],G); od;
gap> for i in [1..8] do m[i]:=List(G_repns,x->ScalarProduct(rho[i],x)); od;
```

The first two lines are simply to initialize the list variables `rho` and `m`. The values of the multiplicities are

```
gap> m[1];m[2];m[3];m[4];m[5];m[6];m[7];m[8];
[ 1, 0, 0, 0, 0, 0, 0, 1, 0 ]
[ 0, 0, 0, 1, 1, 0, 0, 0, 0 ]
[ 0, 0, 0, 0, 0, 0, 0, 0, 1 ]
[ 0, 0, 0, 0, 0, 0, 0, 0, 1 ]
[ 0, 0, 0, 0, 1, 1, 0, 1, 0 ]
[ 0, 1, 0, 0, 0, 0, 1, 0, 1 ]
[ 0, 0, 0, 1, 0, 1, 0, 1, 0 ]
[ 0, 0, 1, 0, 0, 0, 1, 0, 1 ]
```

Only the 3^{rd} and 4^{th} ones are irreducible (recall these are representations induced from a character, so ρ_3, ρ_4 are "principal series" representations of $SL(2,5)$). Which of these induced representaitons ρ_i are trivial on the center? To find out, let us compute, for each representative γ of a conjugacy class $\{\gamma\}$ in G, the class $\{-\gamma\}$. To this end, type

```
gap> for i in [1..Length(G_classes)] do Print(Representative(G_classes[i]),"  ",Representative(G_classes_eps[i]),"  ",i,"\n\n"); od;
()  ( 2, 4)( 3, 5)( 6,16)( 7,19)( 8,20)( 9,17)(10,18)(11,21)(12,24)(13,25)
(14,22)(15,23)  1

( 6, 7, 8,10, 9)(11,13,14,12,15)(16,19,20,18,17)(21,25,22,24,23)  ( 2, 4)( 3, 5)
( 6,19, 8,18, 9,16, 7,20,10,17)(11,25,14,24,15,21,13,22,12,23)  2

( 6, 8, 9, 7,10)(11,14,15,13,12)(16,20,17,19,18)(21,22,23,25,24)  ( 2, 4)( 3, 5)
( 6,20, 9,19,10,16, 8,17, 7,18)(11,22,15,25,12,21,14,23,13,24)  3

( 2, 3, 4, 5)( 6,21,16,11)( 7,23,19,15)( 8,24,20,12)( 9,25,17,13)(10,22,18,14)
( 2, 5, 4, 3)( 6,11,16,21)( 7,15,19,23)( 8,12,20,24)( 9,13,17,25)(10,14,18,22)
4

( 2, 4)( 3, 5)( 6,16)( 7,19)( 8,20)( 9,17)(10,18)(11,21)(12,24)(13,25)(14,22)
(15,23)  ()  5

( 2, 4)( 3, 5)( 6,17,10,20, 7,16, 9,18, 8,19)(11,23,12,22,13,21,15,24,14,25)
( 6, 9,10, 8, 7)(11,15,12,14,13)(16,17,18,20,19)(21,23,24,22,25)  6

( 2, 4)( 3, 5)( 6,18, 7,17, 8,16,10,19, 9,20)(11,24,13,23,14,21,12,25,15,22)
( 6,10, 7, 9, 8)(11,12,13,15,14)(16,18,19,17,20)(21,24,25,23,22)  7

( 2, 6, 9, 4,16,17)( 3,11,15, 5,21,23)( 7,14,10,19,22,18)( 8,24,13,20,12,25)
( 2,16, 9)( 3,21,15)( 4, 6,17)( 5,11,23)( 7,22,10)( 8,12,13)(14,18,19)
(20,24,25)  8

( 2, 6,19)( 3,11,25)( 4,16, 7)( 5,21,13)( 8, 9,24)(10,14,15)(12,20,17)
(18,22,23)  ( 2,16,19, 4, 6, 7)( 3,21,25, 5,11,13)( 8,17,24,20, 9,12)
(10,22,15,18,14,23)  9
```

If the conjugacy classes of G are listed (in GAP's ordering) by

$$\{\gamma_1\} = \{1\}, \{\gamma_2\}, ..., \{\gamma_8\},$$

then we have

$\{\gamma\}$	$\{-\gamma\}$
$\{\gamma_1\}$	$\{\gamma_5\}$
$\{\gamma_2\}$	$\{\gamma_6\}$
$\{\gamma_3\}$	$\{\gamma_7\}$
$\{\gamma_4\}$	$\{\gamma_4\}$
$\{\gamma_5\}$	$\{\gamma_1\}$
$\{\gamma_6\}$	$\{\gamma_2\}$
$\{\gamma_7\}$	$\{\gamma_3\}$
$\{\gamma_8\}$	$\{\gamma_9\}$
$\{\gamma_9\}$	$\{\gamma_8\}$

The output suggests that ρ_i is invariant under Z if and only if $\rho_i(\gamma_1) = \rho_i(\gamma_5)$, $\rho_i(\gamma_2) = \rho_i(\gamma_6)$, ... We have

```
gap> ValuesOfClassFunction(rho[1]);
[ 6, 1, 1, 2, 6, 1, 1, 0, 0 ]
gap> ValuesOfClassFunction(rho[2]);
[ 6, 1, 1, -2, 6, 1, 1, 0, 0 ]
gap> ValuesOfClassFunction(rho[3]);
[ 6, 1, 1, 0, -6, -1, -1, 0, 0 ]
gap> ValuesOfClassFunction(rho[4]);
[ 6, 1, 1, 0, -6, -1, -1, 0, 0 ]
gap> ValuesOfClassFunction(rho[5]);
[ 12, E(5)^2+E(5)^3, E(5)+E(5)^4, 0, 12, E(5)^2+E(5)^3, E(5)+E(5)^4, 0, 0 ]
gap> ValuesOfClassFunction(rho[6]);
[ 12, E(5)^2+E(5)^3, E(5)+E(5)^4, 0, -12, -E(5)^2-E(5)^3, -E(5)-E(5)^4, 0, 0 ]
gap> ValuesOfClassFunction(rho[7]);
[ 12, E(5)+E(5)^4, E(5)^2+E(5)^3, 0, 12, E(5)+E(5)^4, E(5)^2+E(5)^3, 0, 0 ]
gap> ValuesOfClassFunction(rho[8]);
[ 12, E(5)+E(5)^4, E(5)^2+E(5)^3, 0, -12, -E(5)-E(5)^4, -E(5)^2-E(5)^3, 0, 0 ]
```

Under this criterion, only $\rho_1, \rho_2, \rho_5, \rho_8$ factor through to $PSL(2, 5)$. However, none of these are irreducible.

8 A_5 as a finitely presented group

The group A_5 has finite presentation [Atlas]

$$< \quad a, b \mid a^2 = b^3 = (ab)^5 = 1 \quad > . \tag{1}$$

To enter this into GAP, type

```
gap> F2 := FreeGroup( "a", "b" );;
gap> G := F2 / [ F2.1^2, F2.2^3, (F2.1*F2.2)^5 ];
<fp group on the generators [ a, b ]>
```

Let us first assume we happen to know that this finitely presented group is isomorphic to A_5. GAP can compute all the surjective homomorphisms $G \to A_5$:

```
gap> hom:=GQuotients(G,A5)[1];
[ a, b ] -> [ (1,4)(2,5), (1,2,3) ]
```

From this A_5 can be "recovered" by typing

```
gap> A5_fp := Image( hom );
Group([ (1,4)(2,5), (1,2,3) ])
gap> Elements(A5_fp);
[ (), (3,4,5), (3,5,4), (2,3)(4,5), (2,3,4), (2,3,5), (2,4,3), (2,4,5),
  (2,4)(3,5), (2,5,3), (2,5,4), (2,5)(3,4), (1,2)(4,5), (1,2)(3,4), (1,2)(3,5),
  (1,2,3), (1,2,3,4,5), (1,2,3,5,4), (1,2,4,5,3), (1,2,4), (1,2,4,3,5),
  (1,2,5,4,3), (1,2,5), (1,2,5,3,4), (1,3,2), (1,3,4,5,2), (1,3,5,4,2),
  (1,3)(4,5), (1,3,4), (1,3,5), (1,3)(2,4), (1,3,2,4,5), (1,3,5,2,4),
  (1,3)(2,5), (1,3,2,5,4), (1,3,4,2,5), (1,4,5,3,2), (1,4,2), (1,4,3,5,2),
  (1,4,3), (1,4,5), (1,4)(3,5), (1,4,5,2,3), (1,4)(2,3), (1,4,2,3,5),
  (1,4,2,5,3), (1,4,3,2,5), (1,4)(2,5), (1,5,4,3,2), (1,5,2), (1,5,3,4,2),
  (1,5,3), (1,5,4), (1,5)(3,4), (1,5,4,2,3), (1,5)(2,3), (1,5,2,3,4),
  (1,5,2,4,3), (1,5,3,2,4), (1,5)(2,4) ]
```

Since this is a permutation group, you can now carry on the investigation using commands already introduced.

On the other hand, if we had no idea what the group (1) described, we could type instead

```
gap> F2 := FreeGroup( "a", "b" );;
gap> G := F2 / [ F2.1^2, F2.2^3, (F2.1*F2.2)^5 ];
<fp group on the generators [ a, b ]>
gap> iso:=IsomorphismPermGroup(G);
[ a, b ] -> [ (1,2)(4,5), (2,3,4) ]
gap> A5:=Image(iso);
Group([ (1,2)(4,5), (2,3,4) ])
```

As before, since this is a permutation group, you can investige the representations using commands already introduced.

9 Modular representations

This section will be very brief.

If you type

```
gap>p:=7;
gap>A5:=AlternatingGroup(5);
gap>Irr(A5,7);
```

then GAP returns the irreducible characters of the p-modular Brauer table of A_5. If, in this case, $p = 7$ is replaced by a prime $p < 7$ then GAP will return `fail`.

The command below will return the p-modular Brauer character table corresponding to the ordinary character table, where $p = 5$.

```
gap>p:=5;
gap>Display(CharacterTable( "A5" ) mod p);
A5mod5

     2  2  2  .
     3  1  .  1
     5  1  .  .

       1a 2a 3a
    2P 1a 1a 3a
    3P 1a 2a 1a
    5P 1a 2a 3a

X.1      1  1  1
X.2      3 -1  .
X.3      5  1 -1
```

This command works for all primes p.

This brief paper has only given an overview of some of the most basic features of GAP. For more, see [GAP].

References

[Atlas] Robert Wilson, Peter Walsh, Jonathan Tripp, Ibrahim Suleiman, Stephen Rogers, Richard Parker, Simon Norton, Steve Linton and John Bray, **ATLAS of Finite Group Representations**, at `http://www.mat.bham.ac.uk/atlas/html/A5.html`

[GAP] Groups, Algorithms, and Programming (GAP - a free system for computational discrete algebra), at `http://www.gap-system.org/`

[R] J. J. Rotman, **An introduction to the theory of groups**, 4th ed, Springer-Verlag, Grad Texts in Math 148, 1995

[S] J.-P. Serre, **Linear representations of finite groups**, Springer-Verlag, Grad Texts in Math 42, 1977